

An Algorithm for Computing Coefficients of Words in Expressions Involving Exponentials and Its Application to the Construction of Exponential Integrators

Harald Hofstätter

with *W. Auzinger and O. Koch*

Computer Algebra in Scientific Computing (CASC 2019)
August 26–30, 2019, Moscow, Russia

Contents

- ▶ Derivation and implementation of the algorithm
- ▶ Derivation of order conditions for the construction of exponential integrators
- ▶ Non-trivial application: Construction of 8th order commutator-free Magnus-type integrators involving only 8 exponentials

Derivation and implementation of the algorithm

Non-commutative symbols, words

- ▶ \mathcal{A} ... “alphabet” = set of non-commutative symbols, e.g.,

$$\mathcal{A} = \{A, B\} \text{ or } \mathcal{A} = \{A_1, A_2, \dots\}$$

In Maple:

```
> Physics[Setup](noncommutativeprefix = {A, B});
```

- ▶ Word = product of non-commutative symbols
 \mathcal{A}^* ... set of all words over alphabet \mathcal{A} , e.g.,

$$\text{Id}, A, AB, BA, BBB \in \{A, B\}^*$$

Id ... empty word

Expressions involving exponentials

- ▶ Example:

$$X = e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B} - e^{A+B}$$

(relevance of this example will be discussed later)

Exponential function defined by its formal power series

$$e^Y = \exp(Y) = \sum_{n=0}^{\infty} \frac{1}{n!} Y^n$$

- ▶ Formal expansion into a series of words:

$$X = \sum_{w \in \mathcal{A}^*} c_w w$$

$X \in \mathbb{C}\langle\langle \mathcal{A} \rangle\rangle$... algebra of formal power series in the non-commutative variables $\in \mathcal{A}$

- ▶ How calculate coefficients $c_w = \text{coeff}(w, X)$, $w \in \mathcal{A}^*$?

A family of homomorphisms

- ▶ For each word $w = w_1 \cdots w_{\ell(w)} \in \mathcal{A}^*$ of length $\ell(w)$ define map

$$\varphi_w : \mathbb{C}\langle\langle \mathcal{A} \rangle\rangle \rightarrow \mathbb{C}^{(\ell(w)+1) \times (\ell(w)+1)}$$

by

$$\varphi_w(X)_{i,j} = \begin{cases} \text{coeff}(w_{i:j-1}, X), & \text{if } i < j, \\ \text{coeff}(\text{Id}, X), & \text{if } i = j, \\ 0, & \text{if } i > j \end{cases}$$

Here, $w_{i:j-1} = w_i w_{i+1} \cdots w_{j-1}$ = subword of w of length $j - i$, starting at position i and ending at position $j - 1$

- ▶ φ_w is algebra homomorphism:

$$\varphi_w(\alpha X + \beta Y) = \alpha \varphi_w(X) + \beta \varphi_w(Y), \quad X, Y \in \mathbb{C}\langle\langle \mathcal{A} \rangle\rangle, \alpha, \beta \in \mathbb{C}$$

$$\varphi_w(X \cdot Y) = \varphi_w(X) \cdot \varphi_w(Y), \quad X, Y \in \mathbb{C}\langle\langle \mathcal{A} \rangle\rangle$$

- ▶ Furthermore, if $\text{coeff}(\text{Id}, X) = 0$, then

$$\varphi_w(\exp X) = \exp \varphi_w(X)$$

Here, $\varphi_w(X) \dots$ strictly upper triangular \Rightarrow nilpotent \Rightarrow

Its exponential exactly computable in a finite number of steps

A family of homomorphisms, example

$$\begin{aligned}
 \varphi_{AAB}(e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B}) &= \exp\left(\frac{1}{2}\varphi_{AAB}(B)\right) \cdot \exp(\varphi_{AAB}(A)) \cdot \exp\left(\frac{1}{2}\varphi_{AAB}(B)\right) \\
 &= \exp\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{2} \\ 0 & 0 & 0 & 0 \end{pmatrix} \cdot \exp\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \cdot \exp\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{2} \\ 0 & 0 & 0 & 0 \end{pmatrix} \\
 &= \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & \frac{1}{2} \\ 0 & 0 & 0 & 1 \end{pmatrix} \cdot \begin{pmatrix} 1 & 1 & \frac{1}{2} & 0 \\ 0 & 1 & 1 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix} \cdot \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & \frac{1}{2} \\ 0 & 0 & 0 & 1 \end{pmatrix} \\
 &= \begin{pmatrix} 1 & 1 & \frac{1}{2} & \frac{1}{4} \\ 0 & 1 & 1 & \frac{1}{2} \\ 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 1 \end{pmatrix} = \begin{pmatrix} c_{\text{Id}} & c_A & c_{AA} & c_{AAB} \\ 0 & c_{\text{Id}} & c_A & c_{AB} \\ 0 & 0 & c_{\text{Id}} & c_B \\ 0 & 0 & 0 & c_{\text{Id}} \end{pmatrix}
 \end{aligned}$$

Similar calculations involving $\varphi_w(e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B})$ for $w \in \{\text{AAA}, \text{AAB}, \text{ABA}, \text{BAA}, \text{ABB}, \text{BAB}, \text{BBA}, \text{BBB}\}$ yield

$$\begin{aligned}
 e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B} &= \text{Id} + A + B + \frac{1}{2}AA + \frac{1}{2}AB + \frac{1}{2}BA + \frac{1}{2}BB \\
 &\quad + \frac{1}{6}AAA + \frac{1}{4}AAB + \frac{1}{4}BAA + \frac{1}{8}ABB + \frac{1}{4}BAB + \frac{1}{8}BBA + \frac{1}{6}BBB + \dots,
 \end{aligned}$$

where dots represent terms involving words of length > 3

A family of homomorphisms, example

Similarly, calculations like

$$\begin{aligned}\varphi_{\text{AAB}}(e^{\text{A+B}}) &= \exp(\varphi_{\text{AAB}}(\text{A}) + \varphi_{\text{AAB}}(\text{B})) \\ &= \exp\left(\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 \end{pmatrix}\right) = \begin{pmatrix} 1 & 1 & \frac{1}{2} & \frac{1}{6} \\ 0 & 1 & 1 & \frac{1}{2} \\ 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 1 \end{pmatrix}\end{aligned}$$

lead to

$$\begin{aligned}e^{\text{A+B}} &= \text{Id} + \text{A} + \text{B} + \frac{1}{2}\text{AA} + \frac{1}{2}\text{AB} + \frac{1}{2}\text{BA} + \frac{1}{2}\text{BB} \\ &\quad + \frac{1}{6}\text{AAA} + \frac{1}{6}\text{AAB} + \frac{1}{6}\text{ABA} + \frac{1}{6}\text{BAA} + \frac{1}{6}\text{ABB} + \frac{1}{6}\text{BAB} + \frac{1}{6}\text{BBA} + \frac{1}{6}\text{BBB} + \dots\end{aligned}$$

and thus

$$\begin{aligned}e^{\frac{1}{2}\text{B}} e^{\text{A}} e^{\frac{1}{2}\text{B}} - e^{\text{A+B}} \\ &= \frac{1}{12}\text{AAB} - \frac{1}{6}\text{ABA} + \frac{1}{12}\text{BAA} - \frac{1}{24}\text{ABB} + \frac{1}{12}\text{BAB} - \frac{1}{24}\text{BBA} + \dots \\ &= \frac{1}{12}[\text{A}, [\text{A}, \text{B}]] - \frac{1}{24}[[\text{A}, \text{B}], \text{B}] + \dots\end{aligned}$$

Here, terms of length < 3 cancel, terms of length 3 can be written as Lie polynomial

Maple implementation

$\text{coeff}(w, X)$ is the entry of $\varphi_w(X)$ in the upper right corner:

$$\text{coeff}(w, X) = \varphi_w(X)_{1, \ell(w)+1}$$

Implement Maple function `phiv` which computes

$$\text{phiv}(w, X, v) = \varphi_w(X) \cdot v \text{ for vector } v \in \mathbb{C}^{\ell(w)+1}$$

without explicitly generating the matrix $\varphi_w(X)$. Then

$$\text{coeff}(w, X) = \text{first component of } \text{phiv}(w, X, (0, \dots, 0, 1)^T)$$

`phiv` recursively traverses expression tree representing expression X
Evaluation branches out depending on whether current node represents

- ▶ a non-commutative symbol (the atomic case which terminates the recursion)
- ▶ a sum of subexpressions
- ▶ a product of subexpressions
- ▶ a power of a subexpression
- ▶ a commutator of subexpressions, or
- ▶ an exponential of a subexpression

Maple implementation

```
> phiv := proc (w, X, v)
  local i, v1, v2, f, zero;
  if type(X, name) and type(X, noncommutative) then
    return [seq('if'(op(i,w)=X, v[i+1], 0),
                i=1..nops(w)), 0]
  elif type(X, '+') then
    return add(phiv(w, op(i, X), v), i=1..nops(X))
  elif type(X, '*') then
    v1 := v; zero := [0$nops(w)+1];
    for i from nops(X) to 1 by -1 do
      v1 := phiv(w, op(i, X), v1);
      if v1=zero then return zero end if;
    end do;
    return v1
  elif type(X, anything^integer) then
    v1 := v; zero := [0$nops(w)+1];
    for i from 1 to op(2, X) do
      v1 := phiv(w, op(1, X), v1);
      if v1=zero then return zero end if;
    end do;
    return v1
  elif (type(X, function) and
        op(0, X) = Physics[Commutator]) then
    return phiv(w, op(1, X), phiv(w, op(2, X), v))
      - phiv(w, op(2, X), phiv(w, op(1, X), v))
  elif type(X, exp(anything)) then
    v1 := v; v2 := v; zero := [0$nops(w)+1]; f := 1;
    for i from 1 to nops(w) do
      f := f*i; v1 := phiv(w, op(X), v1);
      if v1=zero then return v2 end if;
      v2 := v2 + v1/f;
    end do;
    return v2
  end if;
  return [seq(X*x, x=v)]
end proc;
```

```
> wcoeff := proc (w, X)
  return phiv(w, X, [0$nops(w), 1])[1]
end proc;
```

Derivation of order conditions for the construction of exponential integrators

Exponential integrators

- ▶ Example: evolution equation of the form

$$\partial_t u(t) = Au(t) + Bu(t)$$

We already obtained

$$e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B} - e^{A+B} = \frac{1}{12}[A, [A, B]] - \frac{1}{24}[[A, B], B] + \dots$$

Substitute $A \simeq \tau A = O(\tau)$, $B \simeq \tau B = O(\tau) \rightsquigarrow$

$$e^{\frac{1}{2}\tau B} e^{\tau A} e^{\frac{1}{2}\tau B} = e^{\tau(A+B)} + O(\tau^3)$$

\rightsquigarrow 2nd order Strang splitting; $\tau \dots$ *step-size*

- ▶ More generally, consider expressions of the form

$$X = e^{\Phi_J} \dots e^{\Phi_1} - e^{\Omega}$$

Φ_1, \dots, Φ_J , and $\Omega \dots$ linear combinations of non-commutative symbols $\in \mathcal{A}$ and commutators thereof (=Lie polynomials)

$e^{\Phi_J} \dots e^{\Phi_1} \dots$ exponential integrator

$e^{\Omega} \dots$ exact local solution operator

$X \dots$ local error

Grading of words and homogeneous Lie elements, free graded Lie algebra

- ▶ Symbols $\in \mathcal{A}$ represent objects depending on parameter τ
Choose grading function on \mathcal{A}

$$\text{grade}(a) \in \{1, 2, \dots\}, \quad a \in \mathcal{A}$$

such that it reflects order of magnitude of represented objects:

$$a \simeq O(\tau^{\text{grade}(a)}), \quad a \in \mathcal{A}$$

- ▶ Extension to words:

$$\text{grade}(w) = \sum_{j=1}^{\ell(w)} \text{grade}(w_j), \quad w = w_1 \dots w_{\ell(w)} \in \mathcal{A}^*$$

- ▶ Lie element is called *homogeneous Lie element of grade q* if it can be expanded into linear combination of words of grade q
 \rightsquigarrow free graded Lie algebra generated by \mathcal{A} :

$$[\mathbb{C}\langle \mathcal{A} \rangle] = \bigoplus_{q=1}^{\infty} \mathfrak{g}_q, \quad \mathfrak{g}_q = \{\text{homogeneous Lie elements of grade } q\}$$

Leading error term

- ▶ **Theorem:** $\Phi_1, \dots, \Phi_J, \Omega \dots$ Lie polynomials in variables \mathcal{A}
Expansion into series of words:

$$e^{\Phi_J} \dots e^{\Phi_1} - e^{\Omega} = \sum_{w \in \mathcal{A}^*} c_w w = \Theta + R$$

q_{\min} ... minimal q such that \exists terms $c_w w$, $c_w \neq 0$ of grade q
 Θ contains terms of minimal grade q_{\min}

R contains terms of grade $> q_{\min}$

Then: Θ representing the *leading error term* is a *homogeneous Lie element of grade q_{\min}*

- ▶ $e^{\Phi_J} \dots e^{\Phi_1}$ *self-adjoint* (symmetric) if for

$$\Phi_j = \sum_k X_{j,k}, \quad X_{j,k} \in \mathfrak{g}_k : \quad \Phi_{J-j+1} = \sum_k (-1)^{k+1} X_{j,k}$$

Conforms with the usual definition of an self-adjoint exponential integrator

Lemma: $e^{\Phi_J} \dots e^{\Phi_1}$ and e^{Ω} self-adjoint $\Rightarrow q_{\min}$ odd

Leading error term (representation in a basis)

- ▶ Representation of homogeneous Lie element of grade q in a basis \mathcal{B}_q of \mathfrak{g}_q :

$$\Theta = \sum_{b \in \mathcal{B}_q} c_b b$$

How compute coefficients c_b ?

- ▶ \mathcal{W}_q ... set of words of grade q such that matrix

$$T_q = (\text{coeff}(w, b))_{w \in \mathcal{W}_q, b \in \mathcal{B}_q}$$

is invertible

- ▶ From $c_w = \text{coeff}(w, \Theta) = \sum_{b \in \mathcal{B}_q} c_b \text{coeff}(w, b)$ for $w \in \mathcal{W}_q$ it follows

$$(c_w)_{w \in \mathcal{W}_q} = T_q \cdot (c_b)_{b \in \mathcal{B}_q}$$

and thus

$$(c_b)_{b \in \mathcal{B}_q} = T_q^{-1} \cdot (c_w)_{w \in \mathcal{W}_q}$$

- ▶ Suitable choices for $\mathcal{W}_q, \mathcal{B}_q$: *Lyndon words* of grade q and corresponding *Lyndon basis*

Lyndon words and Lyndon bases

\mathcal{W}_q and \mathcal{B}_q for $\mathcal{A} = \{A, B\}$ and $\text{grade}(A) = \text{grade}(B) = 1$:

q	Lyndon words \mathcal{W}_q	Lyndon basis \mathcal{B}_q
1	A, B	A, B
2	AB	$[A, B]$
3	AAB, ABB	$[A, [A, B]]$, $[[A, B], B]$
4	AAAB, AABB, ABBB	$[A, [A, [A, B]]]$, $[A, [[A, B], B]]$, $[[[A, B], B], B]$
5	AAAAB, AAABB, AABAB, AABBB, ABABB, ABBBB	$[A, [A, [A, [A, B]]]]$, $[A, [A, [[A, B], B]]]$, $[[A, [A, B]], [A, B]]$, $[A, [[[A, B], B], B]]$, $[[A, B], [[A, B], B]]$, $[[[[A, B], B], B], B]$

\mathcal{W}_q and \mathcal{B}_q for $\mathcal{A} = \{A_1, \dots, A_q\}$ and $\text{grade}(A_k) = k$:

q	Lyndon words \mathcal{W}_q	Lyndon basis \mathcal{B}_q
1	A_1	A_1
2	A_2	A_2
3	A_1A_2, A_3	$[A_1, A_2], A_3$
4	$A_1A_1A_2, A_1A_3, A_4$	$[A_1, [A_1, A_2]]$, $[A_1, A_3], A_4$
5	$A_1A_1A_1A_2, A_1A_1A_3, A_1A_2A_2, A_1A_4, A_2A_3, A_5$	$[A_1, [A_1, [A_1, A_2]]]$, $[A_1, [A_1, A_3]]$, $[[A_1, A_2], A_2]$, $[A_1, A_4]$, $[A_2, A_3], A_5$

Order conditions

- ▶ **Theorem:** If for Lie elements $\Phi_1, \dots, \Phi_J, \Omega$ the *order conditions*

$$\text{coeff}(w, e^{\Phi_J} \dots e^{\Phi_1} - e^{\Omega}) = 0, \quad w \in \bigcup_{q=1}^p \mathcal{W}_q$$

are satisfied for all *Lyndon words* of grade $q \leq p$, then these equations are satisfied for *all words* $w \in \mathcal{A}^*$ of grade $\leq p$

- ▶ For $e^{\Phi_J} \dots e^{\Phi_1}$ and e^{Ω} self-adjoint the Theorem holds already if the order conditions are satisfied only for all Lyndon words of *odd* grade $q \leq p$, and we may assume p is odd
- ▶ Recall: symbols $\in \mathcal{A}$ represent objects depending on small parameter τ such that

$$a \simeq O(\tau^{\text{grade}(a)}), \quad a \in \mathcal{A}$$

The conclusion of the Theorem states that

$$e^{\Phi_J} \dots e^{\Phi_1} - e^{\Omega} = O(\tau^{p+1})$$

i.e., $e^{\Phi_J} \dots e^{\Phi_1}$ is an approximation of e^{Ω} of order $p + 1$

Order conditions, example

Wanted: $a, b, c, d \in \mathbb{R}$ such that $e^{bB}e^{aA}e^{cB+d[B,[A,B]]}e^{aA}e^{bB}$ is a 5th order approximation of e^{A+B}

$\mathcal{A} = \{A, B\}$; A, B both have grade 1 corresponding to $A, B \simeq O(\tau)$

Both expressions self-adjoint \rightsquigarrow Lyndon words of odd length ≤ 4 :

$$\mathcal{W} = \mathcal{W}_1 \cup \mathcal{W}_3 = \{A, B, AAB, ABB\}$$

> Physics[Setup] (noncommutativeprefix = {A, B}):

> C := Physics[Commutator]:

> X := exp(b*B)*exp(a*A)*exp(c*B+d*C(B, C(A, B)))*
exp(a*A)*exp(b*B) - exp(A+B):

> W := [[A], [B], [A, A, B], [A, B, B]]:

> eqs := [seq(simplify(wcoeff(w, X)), w in W)];

$$\text{eqs} := \left[-1 + 2a, -1 + 2b + c, -\frac{1}{6} + 2a^2b + \frac{1}{2}a^2c, -\frac{1}{6} + \frac{1}{2}ac^2 + acb + ab^2 - c \right]$$

> sol := solve(eqs);

$$\text{sol} := \left\{ a = \frac{1}{2}, b = \frac{1}{6}, c = \frac{2}{3}, d = \frac{1}{72} \right\}$$

Non-trivial application: Construction of 8th order commutator-free Magnus-type integrators involving only 8 exponentials

Magnus-type integrators

Non-autonomous evolution equation:

$$\partial_t u(t) = A(t)u(t), \quad t \geq t_0, \quad u(t_0) = u_0, \quad A(t) \in \mathbb{C}^{d \times d}$$

One step $(t_n, u_n) \mapsto (t_{n+1}, u_{n+1})$ of step-size τ :

$$t_{n+1} = t_n + \tau, \quad u_{n+1} = \mathcal{S}(\tau, t_n)u_n,$$

where $\mathcal{S}(\tau, t_n) \approx \mathcal{E}(\tau, t_n)$ approximates exact local solution operator

$$\mathcal{E}(\tau, t_n) = e^{\Omega(\tau, t_n)}$$

involving the *Magnus series* $\Omega = \Omega(\tau, t_n)$

Legendre Expansions

Expansion of $A(t_n + t)$ on interval $[t_n, t_n + \tau]$ into a series of shifted Legendre polynomials:

$$A(t_n + t) = A_1 \tilde{P}_0(t) + A_2 \tilde{P}_1(t) + A_3 \tilde{P}_2(t) + \dots, \quad t \in [0, \tau]$$

with

$$\tilde{P}_k(t) = \frac{1}{\tau} P_k\left(\frac{t}{\tau}\right), \quad P_k(x) = (-1)^k \sum_{j=0}^k \binom{k}{j} \binom{k+j}{j} (-1)^j x^j$$

Matrix-valued coefficients A_k given by

$$A_k = (2k - 1)\tau \int_0^1 P_{k-1}(x) A(t_n + \tau x) dx$$

depend on both t_n and τ and satisfy

$$A_k = O(\tau^k)$$

Magnus series in terms of these coefficients:

$$\begin{aligned} \Omega = & A_1 - \frac{1}{6}[A_1, A_2] + \frac{1}{60}[A_1, [A_1, A_3]] - \frac{1}{60}[A_2, [A_1, A_2]] \\ & + \frac{1}{360}[A_1, [A_1, [A_1, A_2]]] - \frac{1}{30}[A_2, A_3] + \dots \end{aligned}$$

Order conditions for Magnus-type integrators

- ▶ Magnus-type integrator: Exponential integrator of the form

$$\mathcal{S}(\tau, t_n) = e^{\tilde{\Phi}_J(\tau, t_n)} \dots e^{\tilde{\Phi}_1(\tau, t_n)}$$

$\tilde{\Phi}_j$... linear combinations of (commutators of) $\tilde{A}_k \approx A_k$
 \tilde{A}_k obtained by numerical approximation of the integral defining A_k

- ▶ $\mathcal{S}(\tau, t_n)$ formally corresponds to

$$S = e^{\Phi_J} \dots e^{\Phi_1}$$

Φ_j ... Lie elements over alphabet $\mathcal{A} = \{A_1, A_2, \dots\}$
with symbols A_k representing $\tilde{A}_k \approx A_k$ and

$$\text{grade}(A_k) = k \text{ conforming to } A_k = O(\tau^k)$$

Order conditions for Magnus-type integrators (continued)

- ▶ Order conditions as in the Theorem:

$$\text{coeff}(w, e^{\Phi_J} \cdots e^{\Phi_1}) - \text{coeff}(w, e^{\Omega}) = 0, \quad w \in \bigcup_{q=1}^p \mathcal{W}_q$$

Have to hold for all *Lyndon words* of grade $\leq p$ over the alphabet $\mathcal{A} = \{A_1, A_2, \dots\}$ with $\text{grade}(A_k) = k$
 \rightsquigarrow Magnus-type integrator of (global) order p
(local order $p + 1 \Rightarrow$ global order p)

- ▶ Explicit formula for $\text{coeff}(w, e^{\Omega})$:

$$\text{coeff}(A_{d_1} \cdots A_{d_\ell}, e^{\Omega}) = \sum_{\substack{(k_1, \dots, k_\ell) \\ 1 \leq k_j \leq d_j}} \prod_{j=1}^{\ell} \frac{(-1)^{d_j+k_j} \binom{d_j-1}{k_j-1} \binom{d_j+k_j-2}{k_j-1}}{\sum_{i=j}^{\ell} k_i}$$

- ▶ $\text{coeff}(w, e^{\Omega}) = 0$ if $\text{grade}(w) \leq p$ and w contains A_d with $d \geq \frac{p}{2} + 1$ (due to orthogonality of Legendre polynomials)

8th order commutator-free Magnus-type integrator

- ▶ Self-adjoint ansatz with 8 exponentials involving A_1, A_2, A_3, A_4 :

> Physics[Setup] (noncommutativeprefix = {A}):

```
> S := exp(f11*A1-f12*A2+f13*A3-f14*A4)*
        exp(f21*A1-f22*A2+f23*A3-f24*A4)*
        exp(f31*A1-f32*A2+f33*A3-f34*A4)*
        exp(f41*A1-f42*A2+f43*A3-f44*A4)*
        exp(f41*A1+f42*A2+f43*A3+f44*A4)*
        exp(f31*A1+f32*A2+f33*A3+f34*A4)*
        exp(f21*A1+f22*A2+f23*A3+f24*A4)*
        exp(f11*A1+f12*A2+f13*A3+f14*A4):
```

- ▶ 22 Lyndon words of odd grade over $\mathcal{A} = \{A_1, A_2, A_3, A_4\}$

\rightsquigarrow 22 order conditions for $p = 8$

BUT: only 16 unknowns f_{11}, \dots, f_{44}

Remarkable that this over-determined system of polynomial equations has solutions

Theoretical explanation for this fact available

8th order commutator-free Magnus-type integrator (cont.)

- ▶ Order conditions corresponding to Lyndon words involving

A_1, A_2 :

```
> W12 := [[A1], [A1, A2], [A1, A1, A1, A2], [A1, A2, A2],  
          [A1, A1, A1, A1, A1, A2], [A1, A1, A1, A2, A2],  
          [A1, A1, A2, A1, A2], [A1, A2, A2, A2]]:
```

```
> rhs12 := [1, -1/6, -1/40, 1/60, -1/1008, 1/420, 1/2520, -1
```

```
> eqs12 := [seq(expand(wcoeff(w, S)), w in W12)] - rhs12:
```

- ▶ Standard Maple function `solve` finds a symbolic representation of the general solution of the system:

```
> vars12 := [f11, f21, f31, f41, f12, f22, f32, f42]:
```

```
> sols12 := solve(eqs12, vars12):
```

- ▶ All possible values in numerical form (very high precision needed!):

```
> Digits := 200:
```

```
> FF := seq(evalf(allvalues(sol)), sol in sols12):
```

99 solutions altogether, 17 real

Each parameter set $f_{11}, f_{21}, f_{31}, f_{41}, f_{12}, f_{22}, f_{32}, f_{42}$ corresponding to any of these solutions can be extended to a full parameter set satisfying the remaining order conditions

8th order commutator-free Magnus-type integrator (cont.)

In

$$S = \prod_{j=8,\dots,1} \exp\left(\sum_{k=1}^4 f_{j,k} A_k\right)$$

substitute

$$A_k \simeq (2k-1)\tau \int_0^1 P_{k-1}(x) A(t_n + \tau x) dx$$

by

$$(2k-1)\tau \sum_{l=1}^K w_k P_{k-1}(x_l) A(t_n + \tau x_l)$$

with Gaussian quadrature nodes and weights of order 8

$$(x_k) = \left(\frac{1}{2} - \sqrt{\frac{15+2\sqrt{30}}{140}}, \frac{1}{2} - \sqrt{\frac{15-2\sqrt{30}}{140}}, \frac{1}{2} + \sqrt{\frac{15-2\sqrt{30}}{140}}, \frac{1}{2} + \sqrt{\frac{15+2\sqrt{30}}{140}} \right),$$

$$(w_k) = \left(\frac{1}{4} - \frac{\sqrt{30}}{72}, \frac{1}{4} + \frac{\sqrt{30}}{72}, \frac{1}{4} + \frac{\sqrt{30}}{72}, \frac{1}{4} - \frac{\sqrt{30}}{72} \right)$$

8th order commutator-free Magnus-type integrator (cont.)

We obtain 8th order commutator-free Magnus-type integrator involving 8 exponentials:

$$\mathcal{S}(t_n, \tau) = \prod_{j=8, \dots, 1} \exp\left(\tau \sum_{k=1}^4 a_{j,k} A(t_n + \tau x_k)\right)$$

with coefficients $a_{j,k}$ given by:

$k = 1$	$k = 2$	$k = 3$	$k = 4$
-1.232611007291861933e+0	1.381999278877963415e-1	-3.352921035850962622e-2	6.861942424401394962e-3
1.452637092757343214e+0	-1.632549976033022450e-1	3.986114827352239259e-2	-8.211316003097062961e-3
-1.783965547974815151e-2	-8.850494961553933912e-2	-1.299159096777419811e-2	4.448254906109529464e-3
-2.982838328015747208e-2	4.530735723950198008e-1	-6.781322579940055086e-3	-1.529505464262590422e-3
-1.529505464262590422e-3	-6.781322579940055086e-3	4.530735723950198008e-1	-2.982838328015747208e-2
4.448254906109529464e-3	-1.299159096777419811e-2	-8.850494961553933912e-2	-1.783965547974815151e-2
-8.211316003097062961e-3	3.986114827352239259e-2	-1.632549976033022450e-1	1.452637092757343214e+0
6.861942424401394962e-3	-3.352921035850962622e-2	1.381999278877963415e-1	-1.232611007291861933e+0

<https://github.com/HaraldHofstaetter/Expocon.mpl>

Thank you for your attention!